
Introducción

Una macro es una secuencia de comandos guardada o pulsaciones de tecla, almacenados para 
un uso posterior. Un ejemplo de una macro simple es aquella que “escribe” su dirección. El 
lenguaje de macros de LibreOffice es muy flexible, permitiendo la automatización tanto de tareas 
simples como de tareas complejas. Las macros son muy útiles  para repetir la misma tarea una y 
otra vez.

Las macros de LibreOffice se escriben usualmente en un lenguaje llamado LibreOffice Basic, que 
algunas veces es abreviado como Basic. Aunque puede aprender Basic y escribir macros, escribir 
macros desde cero es una curva muy empinada. El método usual para un principiante es usar la 
grabadora de macros incluida, la cual graba pulsaciones y las guarda para su uso posterior.

La mayoría de las tareas en LibreOffice se llevan a cabo “enviando comandos” los cuales son 
interceptados y usados. La grabadora de macros trabaja grabando los comandos que se envían 
(vea “El dispatch framework” en la página 14).

Su primera macro

Añadir una macro
El primer paso al aprender programación de macros, es encontrar y usar macros existentes. Esta 
sección asume que tiene una macro que quiere utilizar, la cual podría estar en un correo 
electrónico, una página web o incluso en un libro. Para este ejemplo se utiliza la macro en el 
Listado 1. Debe crear una librería y un modulo para contener su macro; para mas información vea 
“Organización de las macros” en la página 15.

Listado 1: Macro simple que dice hola

Sub MacroHola
  Print "Hola"
End Sub

Utilice los siguientes pasos para crear una librería para contener su macro:

 1) Utilice Herramientas > Macros > Organizar Macros > LibreOffice Basic para abrir el 
diálogo de Macros de LibreOffice Basic (Figura 1).

 2) Haga clic en Organizador para abrir el cuadro de dialogo del Organizador de Macros 
LibreOffice Basic (Figura 2) y seleccione la pestaña de Bibliotecas.

 3) Establezca la Ubicación a Mis Macros y Diálogos, la cual es la ubicación por defecto.

 4) Haga clic en Nuevo para abrir el diálogo de Biblioteca Nueva.

 5) Ingrese un nombre de Biblioteca, por ejemplo PruebaBiblioteca y de clic en Aceptar.

 6) Seleccione la pestaña Módulos.

 7) En la lista de Módulo, expanda Mis Macros y seleccione, por ejemplo PruebaBiblioteca. Un
módulo llamado Module1 ya existe y puede contener su macro. Si lo desea, puede hacer 
clic en Nuevo para crear otro módulo en la Biblioteca.

 8) Seleccione Module1 o el nuevo módulo creado y haga clic en el botón editar para abrir el 
Entorno de Desarrollo Integrado (IDE) (Figura 3). El IDE es un editor de texto incluido con 
LibreOffice que permite crear y editar macros.

 9) Cuando se crea un nuevo módulo, este contiene un comentario y una macro vacía llamada
Main, la cual no ejecuta ninguna acción.

 10) Agregue la nueva macro, ya sea antes de Sub Main o después de End Sub. El Listado 2 
muestra que la nueva macro fue agregada antes de Sub Main.

Figura 1: Diálogo de Macros de LibreOffice Basic

Figura 2: Diálogo del Organizador de macros LibreOffice 
Basic

 11) Haga clic en el ícono Compilar  en la barra de herramientas de Macros, para compilar 
la macro.

 12) Para ejecutar HolaMacro en el módulo, coloque el cursor en la subrutina HolaMacro y haga
clic en el ícono Ejecutar Basic  en la barra de herramientas de macros o presione la 
tecla F5. Un pequeño dialogo se abrirá desplegando la palabra “Hola”. Si el cursor no está 
en una subrutina o función se abrirá un diálogo, entonces seleccione la macro a ejecutar.

 13) Haga clic en Aceptar para cerrar este pequeño diálogo.

 14) Para seleccionar y ejecutar cualquier macro en el módulo, haga clic en el ícono 
Seleccionar Macro  en la barra de herramientas Estándar o vaya a Herramientas > 
Macros > Organizar Macros > LibreOffice Basic.

 15) Seleccione la macro y luego haga clic en Ejecutar.

Figura 3: Diálogo del Entorno de Desarrollo Integrado

Listado 2: Module1 después de agregar la nueva macro.

REM  *****  BASIC  *****

Sub HolaMacro
  Print "Hola"
End Sub

Sub Main

End Sub

Grabar una macro
Si tiene que ingresar la misma información repetidamente, puede copiar esta información después
que ha sido ingresada por primera vez, luego pegar la información en su documento cada vez que
desee utilizarla. Sin embargo, si se copia algo en el portapapeles, el contenido del portapapeles 
cambia. Esto significa que tendrá que volver a copiar la información repetida. Para superar este 
problema, puede crear una macro que ingrese su información repetida.

Nota

Para algunos tipos de información que quiere ingresar repetidamente en un 
documento, puede ser conveniente crear un archivo AutoText. Para mas información, 
vea la Guía de Writer Capítulo 3 Trabajando con texto.

Asegúrese de que la grabadora de macros este habilitada yendo a Herramientas > Opciones > 
LibreOffice > Avanzado en la barra de menú principal y seleccionando Activar Grabación de 
macros. Esta característica se encuentra deshabilitada por defecto cuando LibreOffice se instala 
en su computadora.

 1) Para iniciar la grabación de una macro, vaya a Herramientas > Macros > Grabar Macro 
en la barra del menú principal.

 2) Teclee la información deseada o ejecute una serie de operaciones. Por ejemplo, puede 
teclear su nombre.

 3) Para detener la grabación haga clic en el botón Finalizar grabación dentro del pequeño 
dialogo Grabar macro, y una vez hecho esto, se abrirá el diálogo de Macros de LibreOffice 
(Figura 1 en la página 5).

 4) Abra el contenedor de bibliotecas Mis Macros.

 5) Encuentre la biblioteca llamada Standard en Mis Macros. Note que cada contenedor de 
bibliotecas tiene una biblioteca llamada “Standard”.

 6) Seleccione la biblioteca Standard y haga clic en Módulo Nuevo para crear un nuevo 
módulo para contener la macro. Esto abrirá el diálogo de Módulo Nuevo.

 7) Teclee un nombre descriptivo para el nuevo módulo, por ejemplo, grabado, luego haga clic 
en Aceptar para crear el módulo. El diálogo de Macros de LibreOffice ahora muestra el 
nombre del nuevo módulo en la biblioteca Standard.

 8) En el cuadro de texto Nombre de la Macro, teclee un nombre de una macro que usted 
haya grabado, por ejemplo EscribeMiNombre.

 9) Haga clic en Guardar para guardar la macro y cierre el diálogo de Macros de LibreOffice.

Si siguió todos estos pasos, la biblioteca Standard ahora contiene un módulo llamado Grabado y 
este módulo contiene la macro EscribeMiNombre.

Nota

Cuando LibreOffice crea un módulo, automáticamente añade una macro llamada 
Main.

Ejecutar una macro
 1) Vaya a Herramientas > Macros > Ejecutar macro en la barra de menú principal para abrir

el diálogo del selector de macros (Figura 4).

 2) por ejemplo, seleccione su nueva macro EscribeMiNombre y haga clic en Ejecutar.

 3) Alternativamente, puede ir a Herramientas > Macros > Organizar Macros > LibreOffice 
Basic en la barra de menú principal, para abrir el diálogo de macros de LibreOffice Basic, 
seleccione su macro y haga clic en Ejecutar.

Figura 4: Diálogo del selector de macros

Ver y editar macros
Para ver o editar la macro que usted creo:

 1) Vaya a Herramientas > Macros > Organizar Macros > LibreOffice Basic para abrir el 
diálogo de macros de LibreOffice Basic.

 2) Seleccione su nueva macro EscribeMiNombre y haga clic en editar para abrir la macro en 
el IDE de Basic. La macro EscribeMiNombre se muestra en el Listado 3.

La macro EscribeMiNombre no es tan complicada como parece al principio. Aprender unas 
cuantas cosas ayuda significativamente en el entendimiento de las macros. La discusión comienza
con las características en la parte superior de la macro y las describe.

Listado 3: Macro generada “EscribeMiNombre” macro

REM  *****  BASIC  *****
Sub Main

End Sub

sub EscribeMiNombre
rem -------------------------------------------------------------
rem define variables
dim document   as object
dim dispatcher as object
rem -------------------------------------------------------------
rem get access to the document
document   = ThisComponent.CurrentController.Frame

dispatcher = createUnoService("com.sun.star.frame.DispatchHelper")

rem -------------------------------------------------------------
dim args1(0) as new com.sun.star.beans.PropertyValue
args1(0).Name = "Text"
args1(0).Value = "Su nombre"

dispatcher.executeDispatch(document, ".uno:InsertText", "", 0, args1())

end sub

Comentar con REM
Todos los comentarios en el código de macro inician con REM, que proviene de la palabra en 
inglés “remark” que significa comentario u observación. Todo el texto que se encuentre después 
de REM y que se encuentre en la misma linea se ignora. El apóstrofo (‘) también puede se puede 
usar como un atajo para iniciar un comentario.

LibreOffice no distingue mayúsculas de minúsculas en las palabras clave, por lo que REM, Rem y 
rem pueden todos iniciar un comentario. Si utiliza constantes simbólicas definidas por la Interfaz 
de Programación de Aplicaciones (por sus siglas en Inglés, API), es mas seguro asumir que en los
nombres se discrimina de mayúsculas y minúsculas. Las constantes simbólicas son un tema 
avanzado que no está cubierto por esta guía y no se requieren cuando se usa la grabadora de 
macros en LibreOffice.

Definir subrutinas con SUB
Las macros individuales son almacenadas en subrutinas y estas subrutinas comienzan con la 
palabra clave SUB. El final de una subrutina esta indicado por las palabras END SUB. El código 
comienza por una subrutina llamada Main, que está vacía y que no hace nada. La siguiente 
subrutina, EscribeMiNombre, contiene el código generado por su macro.

Note

LibreOffice siempre crea una subrutina vacía llamada Main cuando crea un módulo.

Existen temas avanzados que están fuera del alcance de esta guía, pero saber de ellos tal vez 
sea de su interés:

• Puede escribir una macro para que los valores puedan pasar a la subrutina. Estos valores 
son llamados argumentos. Sin embargo, las macros grabadas no aceptan argumentos.

• Otra clase de subrutina es llamada función, la cual es una subrutina que devuelve un valor.
Las funciones se definen con la palabra clave FUNCTION en el inicio. Sin embargo, las 
macros grabadas siempre crean subrutinas, no funciones.

Definir variables usando Dim
Puede escribir información en un pedazo de papel para que pueda buscarla después. Una 
variable, como una pieza de papel, contiene información que se puede cambiar y leer. La palabra 
clave Dim originalmente proviene de Dimensión y fue usada para definir las dimensiones de una 
matriz. La instrucción Dim usada en la macro EscribeMiNombre es similar a dejar a un lado un 
pedazo de papel que se usará para almacenar un mensaje o una nota.

En la macro EscribeMiNombre, las variables document y dispatcher se definen como del tipo 
object. Otros tipos de variables comunes incluyen string, integer y date. Una tercera variable, 
llamada args1, es una matriz de valores de propiedad. Una variable de tipo array, permite que una
sola variable contenga valores múltiples, similar a almacenar múltiples páginas en un solo libro. 
Los valores en una matriz usualmente se numeran comenzando desde cero. El número entre 

paréntesis indica el número más alto utilizable para acceder a una ubicación de almacenamiento. 
En este ejemplo, solo hay un valor y está numerado como cero.

Explicando el código de la macro
La siguiente es una explicación del código usado en la macro EscribeMiNombre. Puede que no 
entienda todos los detalles, pero la explicación de cada linea de código puede darle una idea de 
como trabaja la macro.

sub EscribeMiNombre

Define el inicio de la macro

dim document   as object

Define document como una variable. Los object son un tipo de variable específica con 
múltiples campos que representan propiedades y acciones.

dim dispatcher as object

Define dispatcher como una variable de tipo object.

document = ThisComponent.CurrentController.Frame

ThisComponent se refiere al documento actual.

CurrentController es una propiedad que hace referencia a un servicio que controla el 
documento. Por ejemplo, cuando teclea, es CurrentController quien toma nota de lo que 
teclea. Entonces CurrentController envía los cambios al Frame del documento.

Frame es una propiedad del controlador que devuelve el marco principal de un documento. 
Por lo tanto, la variable denominada documento se refiere a la estructura de un documento, 
que recibe los comandos enviados.

dispatcher = createUnoService("com.sun.star.frame.DispatchHelper")

La mayoría de las tareas en LibreOffice se llevan a cabo enviando un comando. LibreOffice 
incluye un servicio de ayuda para el envío (dispatch), el cual hace la mayor parte del trabajo 
cuando se usan dispatches en las macros. El método CreateUnoService acepta el nombre 
de un servicio e intenta crear una instancia de ese servicio. Al completarse, la variable 
dispatcher contiene una referencia a un DispatchHelper.

dim args1(0) as new com.sun.star.beans.PropertyValue

Declara una matriz de propiedades. Cada propiedad tiene un nombre y un valor. En otras 
palabras, es un par de nombre/valor. La matriz creada tiene una propiedad en el índice cero.

args1(0).Name = "Text"

args1(0).Value = "Su nombre"

Da a la propiedad el nombre "Text" y el valor "Su nombre", que es el texto que se inserta 
cuando se ejecuta la macro.

dispatcher.executeDispatch(document, ".uno:InsertText", "", 0, args1())

Aquí es donde ocurre la magia. El servicio de ayuda del envío hace un envío al Frame del 
documento (almacenado en la variable denominada document) con el comando .uno: 
InsertText. Los siguientes dos argumentos, nombre de marco y banderas de búsqueda, 
están fuera del alcance de este documento. El último argumento es la matriz de valores de 
propiedad, que se utilizarán al ejecutar el comando InsertText.

end sub

La última línea del código termina la subrutina.

Crear una macro

Cuando se crea una macro, es importante contestar dos preguntas antes de grabar:

 1) ¿se puede escribir esta tarea como un simple conjunto de comandos?

 2) ¿se pueden ordenar los pasos de manera que el último comando deje el cursor listo para 
el siguiente comando o entrada de texto o datos en el documento?

Un ejemplo más complicado de una macro
Una tarea común es copiar filas y columnas de información desde un sitio web y darles formato 
como tabla en un documento de texto de la siguiente manera:

 1) Copie los datos del sitio web en el portapapeles.

 2) Para evitar fuentes y formato extraños, pegue el texto en un documento de Writer como 
texto sin formato.

 3) Dé formato al texto con tabulaciones entre las columnas de manera que se pueda convertir
en una tabla usando Tabla > Convertir > Texto en tabla en la barra de menú principal.

Con las dos preguntas anteriores en mente, inspeccione el texto para ver si se puede grabar una 
macro para dar formato al texto. En la Figura 5 se muestra un ejemplo de información copiada 
mostrando el grupo de constantes de grosor de letras del sitio web de la API. La primera columna 
en este ejemplo, indica un nombre de constante y cada nombre es seguido por un espacio y una 
tabulación, y cada línea tiene dos espacios al final.

La primera columna en la tabla debe contener un valor numérico, la segunda columna un nombre 
y la tercera columna la descripción. Esta conversión se logra fácilmente para cada fila excepto 
para DONTKNOW y NORMAL, los cuales no contienen un valor numérico, pero los valores están 
entre 1 y 100 y pueden ser ingresados manualmente.

DONTKNOW El grosor de fuente no está especificado. 

THIN especifica un 50% de grosor de fuente.  

ULTRALIGHT especifica un 60% de grosor de fuente.  

LIGHT especifica un 75% de grosor de fuente.  

SEMILIGHT especifica un 90% de grosor de fuente.  

NORMAL especifica un grosor de fuente normal.  

SEMIBOLD especifica un 110% de grosor de fuente.  

BOLD especifica un 150% de grosor de fuente.  

ULTRABOLD especifica un 175% de grosor de fuente.  

BLACK especifica un 200% de grosor de fuente.  

Figura 5: Ejemplo de información copiada

Los datos se pueden limpiar de muchas formas, todas ellas fáciles de lograr. El ejemplo mostrado 
a continuación, usa teclas asumiendo que el cursor se encuentra en el inicio de la línea con el 
texto THIN.

 1) Asegúrese de que la grabación de macros está habilitada yendo, en la barra de menú 
principal, a Herramientas > Opciones > LibreOffice > Avanzado y seleccionando la 
opción Activar grabación de macros. Por defecto, esta característica esta deshabilitada 
cuando se instala LibreOffice en su computadora.

 2) Para iniciar la grabación, vaya a Herramientas > Macros > Grabar macro en la barra de 
menú principal.

 3) Presione Ctrl+Flecha derecha para mover el cursor al inicio de “especifica”.

 4) Presione Retroceso dos veces para remover la tabulación y el espacio.

 5) Presione Tab para agregar la tabulación sin el espacio después del nombre de constante

 6) Presione la tecla Supr para borrar la “e” minúscula y luego presione Mayús+E para agregar
una “E” mayúscula.

 7) Presione Ctrl+Flecha derecha dos veces para mover el cursor al inicio del número.

 8) Presione Ctrl+Mayús+Flecha derecha para seleccionar y mover el cursor después del 
signo %.

 9) Presione Ctrl+C para copiar el texto seleccionado al portapapeles.

 10) Presione la tecla Fin para mover el cursor al final de la línea.

 11) Presione Retroceso dos veces para remover los dos espacios al final.

 12) Presione Inicio para mover el cursor al inicio de la línea.

 13) Presione Ctrl+V para pegar el número seleccionado al inicio de la línea.

 14) Al pegar el valor, también se pega un espacio extra, así que presione Retroceso para 
remover este espacio adicional.

 15) Presione Tab para insertar una tabulación entre el número y el nombre.

 16) Presione Inicio para mover el cursor al inicio de la línea.

 17) Presione Flecha abajo para mover el cursor a la siguiente línea.

 18) Detenga la grabación de la macro y guárdela, vea “Grabar una macro” en la página 6.

Toma mucho mas tiempo leer y escribir los pasos que grabar la macro. Trabaje lentamente y 
piense sobre los pasos mientras los ejecuta. Con práctica se convierte en algo natural.

El código de la macro generada en el Listado 4 se ha modificado para contener el número del 
paso en los comentarios para que el código coincida con los pasos anteriores.

Listado 4: Copiando valores numéricos al inicio de la columna

sub CopiarNumACol1
rem -------------------------------------------------------------
rem define variables
dim document   as object
dim dispatcher as object
rem -------------------------------------------------------------
rem get access to the document
document   = ThisComponent.CurrentController.Frame
dispatcher = createUnoService("com.sun.star.frame.DispatchHelper")

rem (3) Presione Ctrl+Flecha derecha para mover el cursor al inicio de 
“especifica”.
dispatcher.executeDispatch(document, ".uno:GoToNextWord", "", 0, Array())

rem (4) Presione Retroceso dos veces para remover la tabulación y el espacio.
dispatcher.executeDispatch(document, ".uno:SwBackspace", "", 0, Array())

rem -------------------------------------------------------------
dispatcher.executeDispatch(document, ".uno:SwBackspace", "", 0, Array())

rem (5) Presione Tab para agregar la tabulación sin el espacio después del 
nombre de constante.
dim args4(0) as new com.sun.star.beans.PropertyValue

args4(0).Name = "Text"
args4(0).Value = CHR$(9)

dispatcher.executeDispatch(document, ".uno:InsertText", "", 0, args4())

rem (6) Presione la tecla Supr para borrar la “e” minúscula....
dispatcher.executeDispatch(document, ".uno:Delete", "", 0, Array())

rem (6) ... y luego presione Mayús+E para agregar una “E” mayúscula.
dim args6(0) as new com.sun.star.beans.PropertyValue
args6(0).Name = "Text"
args6(0).Value = "E"

dispatcher.executeDispatch(document, ".uno:InsertText", "", 0, args6())

rem (7) Presione Ctrl+Flecha derecha dos veces para mover el cursor al inicio 
del número.
dispatcher.executeDispatch(document, ".uno:GoToNextWord", "", 0, Array())

rem -------------------------------------------------------------
dispatcher.executeDispatch(document, ".uno:GoToNextWord", "", 0, Array())

rem (8) Presione Ctrl+Mayús+Flecha derecha para seleccionar y mover el cursor 
después del signo %.
dispatcher.executeDispatch(document, ".uno:WordRightSel", "", 0, Array())

rem (9) Presione Ctrl+C para copiar el texto seleccionado al portapapeles.
dispatcher.executeDispatch(document, ".uno:Copy", "", 0, Array())

rem (10) Presione la tecla Fin para mover el cursor al final de la línea.
dispatcher.executeDispatch(document, ".uno:GoToEndOfLine", "", 0, Array())

rem (11) Presione Retroceso dos veces para remover los dos espacios al final.
dispatcher.executeDispatch(document, ".uno:SwBackspace", "", 0, Array())

rem -------------------------------------------------------------
dispatcher.executeDispatch(document, ".uno:SwBackspace", "", 0, Array())

rem (12) Presione Inicio para mover el cursor al inicio de la línea.
dispatcher.executeDispatch(document, ".uno:GoToStartOfLine", "", 0, Array())

rem (13) Presione Ctrl+V para pegar el número seleccionado al inicio de la 
línea.
dispatcher.executeDispatch(document, ".uno:Paste", "", 0, Array())

rem (14) Presione Retroceso para remover este espacio adicional .
dispatcher.executeDispatch(document, ".uno:SwBackspace", "", 0, Array())

rem (15) Presione Tab para insertar una tabulación entre el número y el nombre.
dim args17(0) as new com.sun.star.beans.PropertyValue
args17(0).Name = "Text"
args17(0).Value = CHR$(9)

dispatcher.executeDispatch(document, ".uno:InsertText", "", 0, args17())
rem (16) Presione Inicio para mover el cursor al inicio de la línea.
dispatcher.executeDispatch(document, ".uno:GoToStartOfLine", "", 0, Array())

rem (17) Presione Flecha abajo para mover el cursor a la siguiente línea.
dim args19(1) as new com.sun.star.beans.PropertyValue
args19(0).Name = "Count"
args19(0).Value = 1
args19(1).Name = "Select"
args19(1).Value = false

dispatcher.executeDispatch(document, ".uno:GoDown", "", 0, args19())
end sub

Los movimientos del cursor se utilizan para todas las operaciones (contrario a hacer una 
búsqueda). Si ejecuta la macro en la línea de DONTKNOW, la palabra “de” se mueve al inicio de 
la línea y la palabra “El” al inicio de la línea parecería que no sufre ningún cambio. Esto no es 
perfecto, pero no debe ejecutar la macro en las líneas que no tienen el formato apropiado, en 
estas tiene que hacer los cambios manualmente.

Ejecutar una macro rápidamente
Es tedioso ejecutar la macro repetidamente usando Herramientas > Macros > Ejecutar macro 
en la barra de menú principal, cuando la macro puede ser ejecutada desde la IDE (Figura 3 en la 
página 6).

 1) Vaya a Herramientas > Organizar macros > LibreOffice Basic en la barra de menú 
principal para abrir el diálogo de macros de Basic (Figura 1 en la página 5).

 2) Seleccione su macro y haga clic en Editar para abrir la macro en la IDE.

 3) Haga clic en el ícono Ejecutar Basic  en la barra de herramientas de Macros o 
presione la tecla F5 para ejecutar la macro.

 4) A menos que haya modificado la primera macro, esta será una macro vacía llamada Main. 
Modifique Main para que se lea como se muestra en el Listado 5.

 5) Ahora, puede ejecutar CopiarNumACol1 haciendo clic repetidamente en el ícono Ejecutar 
Basic en la barra de herramientas de la IDE. Esto es muy rápido y fácil, especialmente 
para macros temporales que se van a usar unas cuantas veces antes de ser descartadas.

Listado 5: Modificación de Main para llamar a CopiarNumACol1.

Sub Main
  CopiarNumACol1
End Sub

Fallas de la grabadora de macros

Algunas veces la grabadora de macros falla y entender la operación interna de LibreOffice ayuda 
a entender cómo y por qué la grabadora de macros falla algunas veces. El principal culpable está 
relacionado con el dispatch framework (marco de envío) y su relación con la grabadora de macros.

El dispatch framework
El propósito del dispatch framework es proveer acceso uniforme a los componentes (documentos) 
para los comandos que normalmente corresponden a elementos de menú. Puedo usar Archivo > 
Guardar en el menú, el atajo de teclado Ctrl+G o hacer clic en el icono Guardar de la barra de 
herramientas. Todos estos comandos son traducidos en el mismo “dispatch command”.

El dispatch framework se puede usar también para enviar comandos de vuelta a la interfaz de 
usuario (IU). Por ejemplo, después de guardar un nuevo documento, la lista de archivos recientes 
se actualiza.

Un comando dispatch es texto como .uno:InsertObject o .uno:GoToStartOfLine. El comando se 
envía al frame del documento y este pasa el comando hasta que encuentra un objeto que puede 
manejar el comando.

Cómo usa la grabadora de macros el dispatch framework
La grabadora de macros graba los dispatches generados. La grabadora es una herramienta 
relativamente sencilla de implementar y el mismo comando que se envía se graba para uso 
posterior. El problema es que no todos los comandos enviados están completos. Por ejemplo, 
insertar un objeto genera el siguiente código:

dispatcher.executeDispatch(document, ".uno:InsertObject", "", 0, Array())

No es posible especificar que clase de objeto va a crear o insertar. Si un objeto se inserta desde 
un archivo, usted no puede especificar qué archivo insertar.

Si durante la grabación de una macro usa Herramientas > Opciones para abrir y modificar 
elementos de configuración. La macro generada no grabará ningún cambio en la configuración, de
hecho, el código generado estará comentado por lo que ni siquiera se ejecutará.

rem dispatcher.executeDispatch(document, ".uno:OptionsTreeDialog", "", 0, Array())

Si se abre un diálogo, es probable que se genere el comando para abrir el diálogo. Generalmente,
cualquier trabajo hecho dentro del diálogo no se graba. Algunos ejemplos incluyen los diálogos de 
organización de macros, insertar caracteres especiales y tipos similares de diálogos. Otros 
problemas posibles usando la grabadora de macros incluyen cosas como insertar una fórmula, 
establecer área de usuario, establecer filtros en Calc, acciones en formularios de bases de datos y
exportar un documento a un archivo PDF cifrado. Nunca se sabe con certeza qué funcionará a 
menos que lo intente. Por ejemplo, las acciones del diálogo Buscar se capturan correctamente.

Otras opciones
Cuando la grabadora de macros no puede resolver un problema específico, la solución habitual es
escribir el código usando objetos de LibreOffice. Desafortunadamente, existe una curva de 
aprendizaje muy pronunciada para estos objetos de LibreOffice. Por lo general es mejor empezar 
por objetos simples e ir diversificando lentamente a medida que vaya aprendiendo. Aprender a 
leer las macros generadas en un buen lugar para empezar.

Si graba macros de Calc, y la grabadora puede generar la macro correctamente, existe un 
complemento que convierte las macros de Calc cuando se graban. El código final manipula 
objetos de LibreOffice en vez de generar dispatches. Esto puede ser muy útil para aprender el 
modelo de objetos y puede descargarse directamente del sitio web:

http://www.paolo-mantovani.org/downloads/DispatchToApiRecorder/

Organización de las macros

En LibreOffice las macros se agrupan en módulos, los módulos se agrupan en bibliotecas y las 
bibliotecas se agrupan en contenedores de bibliotecas. Una biblioteca se usa generalmente como 
una agrupación mayor, ya sea para una categoría de macros o para una aplicación completa. Los 
módulos generalmente dividen la funcionalidad, como la interacción con el usuario o los cálculos. 
Las macros individuales son subrutinas y funciones. La Figura 6 muestra un ejemplo de la 
estructura jerárquica de las bibliotecas de macros en LibreOffice.

Vaya a Herramientas > Macros > Organizar macros > LibreOffice Basic en la barra de menú 
principal para abrir el diálogo de macros de LibreOffice Basic (Figura 1 en la página 5). Todos los 
contenedores de bibliotecas disponibles se muestran en la lista Macro de. Cada documento es un 
contenedor de bibliotecas, capaz de contener múltiples bibliotecas. La aplicación por si misma 
actúa como dos contenedores de bibliotecas, un contenedor para las macros distribuidas con 

LibreOffice llamado Macros de LibreOffice y el otro contenedor para las macros personales 
llamado Mis macros.

Las macros de LibreOffice se almacenan con el código de tiempo de ejecución de la aplicación, el 
cuál no se puede editar a menos que sea administrador. Esto ayuda a proteger las macros pues 
no deberían modificarse y usted no debería almacenar sus propias macros en el contenedor de 
LibreOffice.

A menos que sus macros se apliquen a un único documento y solo a ese documento, sus macros 
deberían seguramente almacenarse en el contenedor Mis macros. El contenedor mis macros se 
almacena en su área de usuario o directorio personal.

Figura 6: Jerarquía de bibliotecas de macros

Si una macro está contenida en un documento, entonces la macro grabada intentará ejecutarse en
ese documento, porque la macro usa principalmente “ThisComponent” para sus acciones.

Cada contenedor de bibliotecas contiene una biblioteca llamada Standard. Es mejor que cree sus 
propias bibliotecas con nombres significativos que usar la biblioteca Standard. No solamente 
porque los nombres significativos son más fáciles de manejar, sino porque que se pueden importar
en otros contenedores de bibliotecas mientras que con la biblioteca Standard esto no es posible.

Precaución
LibreOffice le permite importar bibliotecas en un contenedor de bibliotecas, pero no le 
permite sobrescribir la biblioteca llamada Standard. Por lo tanto, si almacena sus 
macros en la biblioteca Standard no la podrá importar desde otro contenedor de 
bibliotecas.

Del mismo modo que tiene sentido darle a sus bibliotecas nombre significativos, es prudente usar 
nombre significativos para sus módulos. Por defecto, LibreOffice usa nombres como Module1, 
Module2 y así sucesivamente. 

Cuando crea sus macros, debe decidir donde almacenarlas. Almacenar una macro en un 
documento es útil si se va a compartir el documento y quiere que la macro se incluya en el 
documento. Las macros almacenadas se encuentran en el contenedor de bibliotecas llamado Mis 
macros, sin embargo, están disponibles de forma global para todos los documentos.

Las macros no están disponibles hasta que se carga la biblioteca que las contiene. Sin embargo, 
las bibliotecas Standard y Template se cargan automáticamente. Una biblioteca cargada se 
despliega de forma diferente a como lo hace una que no está cargada. Para cargar una biblioteca 
y los módulos que contiene, haga doble clic en la biblioteca.

¿Donde se almacenan las macros?
LibreOffice almacena datos específicos del usuario en un directorio dentro del directorio particular 
del usuario. Vaya a Herramientas > Opciones > LibreOffice > Rutas en la barra de menú 
principal para ver donde se almacenan otros datos de configuración. Por ejemplo, en una 
computadora corriendo Windows XP esta ruta es C:\Documents and Settings\<user 
name>\Application Data. Las macros del usuario están almacenadas en 
LibreOffice\4\user\basic. Cada biblioteca esta almacenada en su propio directorio fuera 
del directorio de Basic.

Para un uso casual, no es necesario entender donde se almacenan las macros. Sin embargo, si 
sabe donde se almacenan puede crear un respaldo, compartirlas o inspeccionarlas si existe un 
error.

Vaya a Herramientas > Macros > Organizar diálogos en la barra del menú principal para abrir el
diálogo del Organizador de macros de LibreOffice (Figura 2 en la página 5). Alternativamente, 
vaya a Herramientas > Macros > Organizar Macros > LibreOffice Basic en la barra de menú 
principal para abrir el diálogo de Macros de LibreOffice (Figura 1 en la página 5) y luego haga clic 
en botón Organizador.

Importar macros
El diálogo del Organizador de macros de LibreOffice le permite importar bibliotecas en su 
documento, así como crear, borrar y renombrar bibliotecas, módulos y diálogos.

 1) Seleccione el contenedor de bibliotecas a utilizar y luego haga clic en Importar para 
importar las bibliotecas de macros (Figura 2 en la página 5).

 2) Navegue al directorio que contiene la biblioteca a importar (Figura 7). Por lo general, hay 
dos archivos para escoger, dialog.xlb y script.xlb. No importa cual de estos archivos 
seleccione, se importarán los dos. Las macros se pueden almacenar en bibliotecas dentro 
de los documentos de LibreOffice. En lugar de seleccionar un directorio en el disco, 
seleccione un documento para importar las bibliotecas contenidas en éste.

Nota  

No puede importar la biblioteca llamada Standard.

Sugerencia
En una computadora operando Linux, los archivos específicos de LibreOffice están 
almacenados en el directorio home del usuario, dentro de un subdirectorio cuyo 
nombre inicia con un punto (generalmente .config/). Los directorios y nombres que 
inicial con un punto pueden estar ocultos y no mostrarse en un diálogo de selección 
normal. Si utiliza los diálogos de LibreOffice en vez de los diálogos específicos del 
sistema operativo, teclee el nombre del directorio deseado en el campo de nombre.

Figura 7: Navegando a la biblioteca de macros

 3) Seleccione un archivo y haga clic en Abrir para continuar, entonces se abrirá el diálogo 
Importar bibliotecas (Figura 8).

Figura 8: Seleccione las opciones 
para importar

 4) Seleccione las siguientes opciones para importar bibliotecas:

a) Si no hay opciones seleccionadas, la biblioteca se copia al directorio de macro del 
usuario. Sin embargo, si la biblioteca que está importando tiene el mismo nombre y la 
está importando a la misma ubicación, entonces no se copiará.

b) Seleccione Reemplazar las bibliotecas existentes si la biblioteca que quiere importar
tiene el mismo nombre y quiere reemplazar la biblioteca existente.

c) Seleccione Insertar como referencia si quiere usar la biblioteca como referencia pero 
no quiere importar la biblioteca en su documento. Cuando una biblioteca se usa como 
referencia permanece en su ubicación y es de solo lectura.

 5) Haga clic en Aceptar para importar la biblioteca de macros que seleccionó.

18 | Importar macros



Descargar macros para importar
Hay macros disponibles para descargar. Algunas macros están contenidas en documentos, otras 
son archivos normales que debe seleccionar e importar y otras son texto de macros para copiar y 
pegar en el IDE de Basic. Vea en “Añadir una macro” en la página 4 cómo agregar macros a su 
bibliotecas de macros y en “Ver y editar macros” en la página 8 cómo editar macros usando la IDE
de Basic.

Algunas macros están disponibles como descargas gratuitas en Internet (vea la Tabla 1). 

Tabla 1. Ejemplos de macros
Localización Descripción

http://www.pitonyak.org/oo.php Material de referencia sobre macros.

http://www.pitonyak.org/database/ 
Material de referencia sobre macros de bases de 
datos.

https://wiki.documentfoundation.org/Macros Muchos enlaces a macros.

http://en.libreofficeforum.org/ 
http://forum.openoffice.org/en/forum/ 

Foros con muchos ejemplos y ayuda.

Cómo ejecutar una macro

Aunque puede usar Herramientas > Macros > Ejecutar Macro para ejecutar todas las macros, 
esto no es eficiente para las macros ejecutadas con mayor frecuencia. Consulte “Ejecutar una 
macro” en la página 7 para más información.

Una técnica para las macros usadas con mayor frecuencia, es enlazar la macro a un icono de 
barra de herramientas, elemento de menú, acceso directo de teclado o un botón incrustado en un 
documento. Al elegir un método, también es bueno hacer preguntas como:

• ¿esta macro debería estar disponible para solo un documento o de forma global para todos
los documentos?

• ¿es esta macro específica para un tipo de documento, como para un documento de Calc?

• ¿con que frecuencia se utilizará la macro?

Las respuestas determinarán dónde almacenar la macro y cómo hacerla disponible. Por ejemplo, 
es probable que no agregue una macro raramente utilizada a una barra de herramientas. Para 
ayudar a determinar sus opciones consulte la Tabla 2.

Tabla 2. Donde almacenar una macro
Tipo de 
macro

LibreOffice (para todos
los componentes)

Un tipo de documento 
específico

Un documento

Barra de 
herramientas

No Sí Sí

Menú No Sí Sí

Atajo de 
teclado

Sí Sí No

Suceso Sí Sí Sí

Barras de herramientas, elementos de menú y atajos de teclado
Para agregar un elemento de menú, un atajo de teclado o un ícono de barra de herramientas que 
llame a una macro, use el diálogo Personalizar (Figura 9). 

Figure 9: Diálogo Personalizar de LibreOffice

El diálogo Personalizar contiene páginas para configurar menús, atajos de teclado, barras de 
herramienta y sucesos. Para abrir este diálogo vaya a Herramientas > Personalizar en la barra 
de menú principal o haga clic con el botón derecho en un espacio vacío de una barra de 
herramientas y seleccione Personalizar barra de herramientas del menú contextual.

La cobertura completa del diálogo Personalizar esta mas allá del alcance de este capítulo. Haga 
clic en el botón Ayuda para entrar en las páginas de ayuda incluidas con LibreOffice o vea el 
Capítulo 14, Personalizar LibreOffice.

Sucesos
Cuando algo sucede en LibreOffice, a esto se le llama suceso o evento. Por ejemplo, abrir un 
documento, presionar una tecla o mover el ratón. LibreOffice permite que los sucesos activen la 
ejecución de macros; la macro entonces es llamada manipulador de sucesos. Una cobertura 
completa de los manipuladores de sucesos está muy por encima del alcance de este documento, 
pero un poco de conocimiento puede ser útil.

Precaución
Sea cuidadoso cuando configure un manipulador de sucesos. Por ejemplo, supongamos 
que escribe un suceso que se llama cada vez que se presiona una tecla, pero comete un 
error y el suceso no se manipula adecuadamente. Un resultado posible es que su 
manipulador de sucesos consumirá todas las pulsaciones del teclado, forzándolo a cerrar 
LibreOffice.

 1) Vaya a Herramientas > Personalizar en la barra de menú principal para abrir el diálogo 
Personalizar y seleccione la pestaña Sucesos (Figura 10). Los sucesos en el diálogo 
Personalizar están relacionados con la aplicación completa o con documentos específicos.

Figura 10: Pestaña de Sucesos en el diálogo Personalizar

 2) En la opción Guardar en, seleccione LibreOffice o un documento específico del menú 
desplegable para guardar su suceso.

 3) Un uso común es asignar el suceso Abrir documento para llamar a una macro específica. 
La macro entonces efectúa ciertas tareas de configuración para el documento. Seleccione 
el suceso deseado y haga clic en Macro para abrir el diálogo Selector de macros (Figura 4 
en la página 8).

 4) Seleccione la macro deseada y haga clic en Aceptar para asignar la macro al suceso. La 
pestaña de sucesos muestra que el suceso ha sido asignado a la macro.

Muchos objetos en un documento se pueden configurar para llamar macros cuando se producen 
sucesos. El uso mas común es para agregar un control, como un botón, en un documento. Incluso
hacer doble clic sobre un gráfico abre un diálogo con una pestaña de macros que permite asignar 
una macro a un suceso..

Extensiones

Una extensión es un paquete que se puede instalar en LibreOffice para añadir una nueva 
funcionalidad. Las extensiones se pueden escribir en casi cualquier lenguaje de programación y 
pueden ser simples o sofisticadas. Las extensiones se pueden agrupar en los tipos:

• Complementos de Calc, los cuales proveen nuevas funcionalidades para Calc, incluyendo 
nuevas funciones que actúan como funciones integradas normales.

• Nuevos componentes y funcionalidad, que normalmente incluye cierto nivel de integración 
con la Interfaz de Usuario (IU) como son nuevos menús o barras de herramientas.

• Tablas dinámicas que son usadas directamente en Calc

Extensiones | 21



• Complementos de Chart con nuevos tipos de gráficos

• Componentes lingüísticos como los correctores ortográficos

• Plantillas de documentos e imágenes

Aunque las extensiones individuales se pueden encontrar en muchos lugares, hay un repositorio 
de extensiones en: http://extensions.libreoffice.org/ y algo de documentación en 
http://libreplanet.org/wiki/Group:OpenOfficeExtensions/List.

Para saber más sobre como obtener e instalar extensiones, vea el Capítulo 14, Personalizar 
LibreOffice.

Escribir macros sin la grabadora

Los ejemplos cubiertos en este capítulo se han creado usando la grabadora de macros y el 
dispatcher. Puede escribir macros que accedan directamente a los objetos comprendidos en 
LibreOffice si se siente confiado de escribir código de programación. En otras palabras, puede 
manipular directamente un documento.

Manipular directamente los objetos internos de LibreOffice es un tema avanzado que va más allá 
del alcance de este capítulo. Sin embargo, un ejemplo sencillo demuestra como funciona esto.

Listado 6: Añadir el texto “Hola” al documento actual.

Sub AnadirHola
  Dim oDoc
  Dim sTextService$
  Dim oCurs
  
  REM ThisComponent se refiere al documento actualmente activo
  oDoc = ThisComponent

  REM Verifica que este es un documento de texto
  sTextService = "com.sun.star.text.TextDocument"
  If NOT oDoc.supportsService(sTextService) Then
    MsgBox "This macro only works with a text document"
    Exit Sub
  End If
  REM obtiene la vista del cursor del controlador actual
  oCurs = oDoc.currentController.getViewCursor()
  
  REM Mueve el cursor al final del documento
  oCurs.gotoEnd(False)
  
  REM Inserta el texto “Hola” al final del documento
  oCurs.Text.insertString(oCurs, "Hello", False)  
End Sub

Encontrar mas información

Existen disponibles numerosos recursos para proveer ayuda en la escritura de macros. Use 
Ayuda > Ayuda de LibreOffice para abrir las páginas de ayuda de LibreOffice. La esquina 
superior izquierda del sistema de ayuda de LibreOffice contiene una lista desplegable que 
determina que conjunto de ayuda se muestra. Para ver la ayuda de Basic, seleccione LibreOffice 
Basic de esta lista.

Material incluido
En LibreOffice se incluyen muchas macros excelentes. Utilice Herramientas > Macros > 
Organizar macros > LibreOffice Basic para abrir el diálogo de macros. Expanda la biblioteca 

Tools en el contenedor de bibliotecas de LibreOffice. Inspeccione el módulo Debug—algunos 
buenos ejemplos incluyen WritedbgInfo(document) y printdbgInfo(sheet).

Recursos Online
Los siguientes enlaces y referencias contienen información acerca de la programación de macros:

http://ask.libreoffice.org/ (un sitio de Q&A donde voluntarios responden a preguntas relacionadas 
con LibreOffice)

http://forum.openoffice.org/en/forum/ (foro de la comunidad de Apache OpenOffice; los voluntarios 
responden preguntas sobre LibreOffice)

http://en.libreofficeforum.org/ (foro no oficial de la comunidad de LibreOffice)

http://api.openoffice.org/docs/common/ref/com/sun/star/module-ix.html (referencia oficial IDL; aquí 
encontrará casí cada comando con una descripción)

https://wiki.documentfoundation.org/Documentation/Other_Documentation_and_Resources (vea 
en la sección de programadores, la Guía de programadores BASIC y la Guía de Desarrolladores; 
este último contiene una explicación detallada)

http://www.pitonyak.org/oo.php (página de macros de Andrew Pitonyak)

http://www.pitonyak.org/AndrewMacro.odt (numerosos ejemplos macros trabajando)

http://www.pitonyak.org/book/ (libro sobre macros de Andrew Pitonyak)

http://www.pitonyak.org/database/ (numerosos ejemplos de macros utilizando Base)

Material impreso y eBooks
Actualmente no hay libros específicos para las macros de LibreOffice que estén disponibles para 
descarga.

La información en los siguientes libros es generalmente aplicable a LibreOffice; los libros están 
disponibles para su compra en formato impreso y eBook de sus editores:

Learn OpenOffice.org Spreadsheet Macro Programming, del Doctor Mark Alexander Bain
Vea http://www.packtpub.com/openoffice-ooobasic-calc-automation/book.

Database Programming with OpenOffice.org Base & Basic, de Roberto Benitez
Vea http://www.lulu.com/product/paperback/database-programming-with-openofficeorg-base-
basic/3568728

